Example 8.10 Heat Transfer in a Rectangle
Equation (8.1.26) is solved in Maple below:

```
> restart:with(inttrans):with(plots):
>
> eq:=diff(u(x,t),t)=diff(u(x,t),x$2);
```

$$eq := \frac{\partial}{\partial t} u(x,t) = \frac{\partial^2}{\partial x^2} u(x,t) \quad (1)$$

```
> u(x,0):=1;
```

$$u(x,0) := 1 \quad (2)$$

```
> bc1:=u(x,t)=0;
```

$$bc1 := u(x,t) = 0 \quad (3)$$

```
> bc2:=u(x,t)=0;
```

$$bc2 := u(x,t) = 0 \quad (4)$$

```
> eqs:=laplace(eq,t,s):
> eqs:=subs(laplace(u(x,t),t,s)=U(x),eqs);
```

$$eqs := s\,U(x) - 1 = \frac{d^2}{dx^2} U(x) \quad (5)$$

```
> bc1:=laplace(bc1,t,s):
> bc1:=subs(laplace(u(x,t),t,s)=U(x),bc1);
```

$$bc1 := U(x) = 0 \quad (6)$$

```
> bc2:=laplace(bc2,t,s):
> bc2:=subs(laplace(u(x,t),t,s)=U(x),bc2);
```

$$bc2 := U(x) = 0 \quad (7)$$

```
> dsolve(eqs,U(x));
```

$$U(x) = e^{\sqrt{s}\,x}\_C2 + e^{-\sqrt{s}\,x}\_C1 + \frac{1}{s} \quad (8)$$

```
> U(x):=c[1]*cosh(s^(1/2)*x)+c[2]*sinh(s^(1/2)*x)+1/s;
```

$$U(x) := c_1 \cosh\left(\sqrt{s}\,x\right) + c_2 \sinh\left(\sqrt{s}\,x\right) + \frac{1}{s} \quad (9)$$

```
> eq0:=eval(subs(x=0,bc1)):
> eq1:=eval(subs(x=1,bc2)):
> con:=solve({eq0,eq1},{c[1],c[2]}):
> U(x):=subs(con,U(x));
```

$$U(x) := -\frac{\cosh\left(\sqrt{s}\,x\right)}{s} + \frac{\left(\cosh\left(\sqrt{s}\right) - 1\right)\sinh\left(\sqrt{s}\,x\right)}{\sinh\left(\sqrt{s}\right)s} + \frac{1}{s} \quad (10)$$

```
> U(x):=factor(combine(simplify(U(x))));
```

(11)

$$U(x) := \frac{\sinh\left(\sqrt{s}\ (x-1)\right) - \sinh\left(\sqrt{s}\ x\right) + \sinh\left(\sqrt{s}\right)}{\sinh\left(\sqrt{s}\right)\ s} \quad (11)$$

```
> P(s):=numer(U(x));
```

$$P(s) := \sinh\left(\sqrt{s}\ (x-1)\right) - \sinh\left(\sqrt{s}\ x\right) + \sinh\left(\sqrt{s}\right) \quad (12)$$

```
> Q(s):=denom(U(x));
```

$$Q(s) := \sinh\left(\sqrt{s}\right)\ s \quad (13)$$

```
> solve(Q(s),s);
```

$$0 \quad (14)$$

```
> _EnvAllSolutions := true;
```

$$\_EnvAllSolutions := true \quad (15)$$

```
> solve(Q(s),s);
```

$$-\pi^2\ \_Z1\sim^2,\ 0 \quad (16)$$

s = 0 is repeated twice (one root coming from 2 and the other coming from the sine term in q(s)). The roots are (where n goes from 1 to infinity):

```
> 0,0,-n^2*Pi^2;
```

$$0,\ 0,\ -n^2\pi^2 \quad (17)$$

```
> mu0:=0;
```

$$\mu 0 := 0 \quad (18)$$

The coefficients B2 and B1 are found as (equation 8.1.22 and 8.1.24):

```
> b[2]:=(s-mu0)^2*P(s)/Q(s);
```

$$b_2 := \frac{s\left(\sinh\left(\sqrt{s}\ (x-1)\right) - \sinh\left(\sqrt{s}\ x\right) + \sinh\left(\sqrt{s}\right)\right)}{\sinh\left(\sqrt{s}\right)} \quad (19)$$

```
> B[2]:=limit(b[2],s=0);
```

$$B_2 := 0 \quad (20)$$

```
> b[1]:=diff(b[2],s):
> B[1]:=limit(b[1],s=0);
```

$$B_1 := 0 \quad (21)$$

For this problem the contribution from the repeated root s = 0 is zero. This is not always true as shown in the next example.

```
> A(s):=P(s)/diff(Q(s),s):
> A[n]:=simplify(subs(s=mu,A(s)));
```

$$A_n := \frac{2\left(\sinh\left(\sqrt{\mu}\ (x-1)\right) - \sinh\left(\sqrt{\mu}\ x\right) + \sinh\left(\sqrt{\mu}\right)\right)}{\cosh\left(\sqrt{\mu}\right)\sqrt{\mu} + 2\sinh\left(\sqrt{\mu}\right)} \quad (22)$$

```
> A[n]:=simplify(subs(mu^(1/2)=I*n*Pi,mu=-n^2*Pi^2,A[n])):
> vars:={cos(n*Pi)=(-1)^n,sin(n*Pi)=0};
```

(23)

$$vars := \{\cos(n\pi) = (-1)^n, \sin(n\pi) = 0\} \quad (23)$$

```
> A[n]:=simplify(subs(vars,A[n])):
> A[n]:=simplify(subs(vars,expand(A[n])));
```

$$A_n := -\frac{2\sin(n\pi x)\left(-1+(-1)^{-n}\right)}{n\pi} \quad (24)$$

```
> b1s:=B[1]*subs(mu0=0,1/(s-mu0));
```

$$b1s := 0 \quad (25)$$

```
> b1t:=invlaplace(b1s,s,t);
```

$$b1t := 0 \quad (26)$$

```
> b2s:=B[2]*subs(mu0=0,1/(s-mu0)^2);
```

$$b2s := 0 \quad (27)$$

```
> b2t:=invlaplace(b2s,s,t);
```

$$b2t := 0 \quad (28)$$

```
> uns:=A[n]/(s-mu);
```

$$uns := -\frac{2\sin(n\pi x)\left(-1+(-1)^{-n}\right)}{n\pi(s-\mu)} \quad (29)$$

```
> unt:=invlaplace(uns,s,t);
```

$$unt := \frac{2\sin(n\pi x)\,e^{\mu t}\left(1+(-1)^{1-n}\right)}{n\pi} \quad (30)$$

```
> unt:=subs(mu=-n^2*Pi^2,unt);
```

$$unt := \frac{2\sin(n\pi x)\,e^{-n^2\pi^2 t}\left(1+(-1)^{1-n}\right)}{n\pi} \quad (31)$$

The solution is obtained and plotted as:

```
> U:=b1t+b2t+Sum(unt,n=1..infinity);
```

$$U := \sum_{n=1}^{\infty} \frac{2\sin(n\pi x)\,e^{-n^2\pi^2 t}\left(1+(-1)^{1-n}\right)}{n\pi} \quad (32)$$

```
> u:=piecewise(t=0,1,t>0,subs(infinity=20,U)):
> plot3d(u,x=0..1,t=0..0.5,axes=boxed,title="Figure Exp. 8.18.",
  labels=[x,t,"u"],orientation=[45,60]);
```

Figure Exp. 8.18.




```
> plot([subs(t=0,u),subs(t=0.01,u),subs(t=0.05,u),subs(t=0.1,u)],
  x=0..1,axes=boxed,title="Figure Exp. 8.19.",thickness=5,labels=
  [x,"u"]);
```

Figure Exp. 8.19.




```
>
```